```
> restart;
> assume(alpha > 1); # shape of container, defining r=alpha^z
> assume(Z::real); # height of container, defining z=-infinity..Z
> the_volume := 473.17647; # desired total volume; 2 cups in cm^3
```

$$the\_volume := 473.17647 \tag{1}$$

```
> ### the_volume := 29.57353; # desired total volume; 2 tablespoons
  in cm^3
> area := simplify(int(2*Pi*alpha^z*sqrt(1+diff(alpha^z,z)^2), z=-
  infinity..Z));
```

$$area := \frac{\pi \left(\alpha\sim^{Z\sim} \sqrt{1 + \ln(\alpha\sim)^2 \alpha\sim^{2 Z\sim}} \ln(\alpha\sim) + \ln\left(\alpha\sim^{Z\sim} \ln(\alpha\sim) + \sqrt{1 + \ln(\alpha\sim)^2 \alpha\sim^{2 Z\sim}} \right) \right)}{\ln(\alpha\sim)^2} \tag{2}$$

```
> volume := simplify(int(Pi*(alpha^z)^2, z=-infinity..Z));
```

$$volume := \frac{1}{2} \frac{\pi \alpha\sim^{2 Z\sim}}{\ln(\alpha\sim)} \tag{3}$$

```
> the_Z := simplify(solve(volume = the_volume, Z));
Warning, solve may be ignoring assumptions on the input
variables.
```

$$the\_Z := \frac{2.85394285 + 0.5 \ln(\ln(\alpha\sim))}{\ln(\alpha\sim)} \tag{4}$$

```
> the_area := simplify(eval(area, Z = the_Z));
```

$$the\_area := \frac{1}{\ln(\alpha\sim)^2} \left(54.52573195 \ln(\alpha\sim)^{3/2} \sqrt{1. + 301.2334965 \ln(\alpha\sim)^3} + 3.141592654 \ln\left(17.35607953 \ln(\alpha\sim)^{3/2} + \sqrt{1. + 301.2334965 \ln(\alpha\sim)^3} \right) \right) \tag{5}$$

```
> opt := Optimization:-Minimize(the_area, alpha=1..10);
```

$$opt := [319.448406338918, [\alpha\sim = 1.19628130817947]] \tag{6}$$

```
> opt_area := op(1,opt);
```

$$opt\_area := 319.448406338918 \tag{7}$$

```
> opt_alpha := eval(alpha, op(2,opt));
```

$$opt\_alpha := 1.19628130817947 \tag{8}$$

```
> opt_Z := eval(the_Z, alpha=opt_alpha);
```

$$opt\_Z := 11.1281682315513 \tag{9}$$

```
> plot3d([opt_alpha^z*cos(theta), opt_alpha^z*sin(theta), z],
  theta=-Pi..Pi, z=opt_Z-log[opt_alpha](4)..opt_Z, scaling=
  constrained); # 15/16 of the_volume
```



11
10
9
8
7
6
5
4
−6
−4
−2
0
2
4
6
6
4
2
0
−2
−4
−6
z